
Test Match NPN TC to Silicon

*       _____
*       |   0|        Use this to simulate Vbe and Beta over temp.
*       |   _|_       Tweek the model terms IS, XTB, and XTI. 
*       |  / _ \      Repeat until simulations match actual silicon.
*      _|_ \/ \/      Different simulators often give different results!
*      /// /\_/\ I1
*          \___/                    V2         R2
*            |                        __/\  /\  /\__
*     _______|_________        C    _|_   \/  \/   _|_
*    |                 |C     ____ /   \           ///
*    |     R1         _|          |EGbas|
*    |___/\  /\  /\_|'  npn   ____|     |
*          \/  \/   |`->       B   \___/
*                  B   |            _|_
*                     _|_            ///
*                     ///
                  
I1      0     C     DC    1e-6
R1      C     B     1
R2      V2    0     1k
Q1      C     B     0     npnv    1
EGbas   V2    0     C     B       1

.control
foreach tempval -55 -35 -15  5 25 45 65 85 105 125
set     temp  = $tempval
op 
let     beta  = mag(1e-6/v2)
let     vbe   = mag(b)
echo    Temp  = $temp C_deg Vbe = $&vbe  Beta = $&beta
end

.endc

.model     npnv         npn ( 
*==========================================================
+IS=1.1E-17   NF=1.0       BF=120      VAF=30      IKF=6e-03    
+             NR=1.0       BR=0.5      VAR=4       IKR=3e-04  
+ISE=9E-17    NE=2                                      
+ISC=1E-21    NC=2                                                 
+RB=750       RBM=450                              IRB=8E-04    
+RE=17        RC=110 
*==========================================================                                          
+CJE=2E-14    VJE=0.65     MJE=0.3                           
+CJC=2E-14    VJC=0.65     MJC=0.3                              
+CJS=3E-14    VJS=0.35     MJS=0.19    XCJC=0.42                 
+TF=2E-11     XTF=1.25     VTF=1       ITF=0.0035               
+TR=6E-09     FC=0.9       PTF=210 
*==========================================================                               
+KF=1.0E-16   AF=1                                                
+XTB=1.4      EG=1.11      XTI=8      TNOM=25 

.end
*********Actual Silicon DATA***********************
**  Temp      NPN_1uA      Beta
**  -55       8.23E-01     67% 
**  -35       7.87E-01   
**  -15       7.43E-01     75%
**    5       6.96E-01
**   25       6.53E-01     100%
**   45       6.09E-01
**   65       5.64E-01     125%
**   85       5.19E-01
**  105       4.74E-01
**  125       4.17E-01     160%
**  tweek     XTI & IS     XTB

=======Circuit: Test Match NPN TC to Silicon========
Temp = -55 C_deg      Vbe = 0.823219      Beta = 62.345
Temp = -35 C_deg      Vbe = 0.782302      Beta = 71.2528
Temp = -15 C_deg      Vbe = 0.740177      Beta = 80.6773
Temp =   5 C_deg      Vbe = 0.696933      Beta = 90.6231
Temp =  25 C_deg      Vbe = 0.652636      Beta = 101.095
Temp =  45 C_deg      Vbe = 0.607372      Beta = 112.1
Temp =  65 C_deg      Vbe = 0.561192      Beta = 123.644
Temp =  85 C_deg      Vbe = 0.514146      Beta = 135.734
Temp = 105 C_deg      Vbe = 0.466281      Beta = 148.378
Temp = 125 C_deg      Vbe = 0.417638      Beta = 161.583


